Problem #1 :

In[∘]:=
```
Clear["Global`*"]
```

Problem 1:
a) the given matrix, M find the eigenvalue and eigenvectors.
b) Normalize the eigenvectors.
Hint: Before you normalize the math should NOT be too complicated.
After you normalize, there is a radical denominator,
but you can factor that out when you check the diagonalization properties.

$$\begin{pmatrix} 7 & 3 \\ 3 & -1 \end{pmatrix}$$

In[∘]:=
```
m = {{7, 3}, {3, -1}};
m // MatrixForm
```

Out[∘]//MatrixForm=

$$\begin{pmatrix} 7 & 3 \\ 3 & -1 \end{pmatrix}$$

In[∘]:=
```
Eigenvalues[m]
```

Out[∘]= {8, -2}

In[∘]:=
```
ev = Normalize /@ Eigenvectors[m]
```

Out[∘]= $\left\{\left\{\frac{3}{\sqrt{10}}, \frac{1}{\sqrt{10}}\right\}, \left\{-\frac{1}{\sqrt{10}}, \frac{3}{\sqrt{10}}\right\}\right\}$

In[∘]:=
```
ev.m.Transpose[ev] // Simplify
```

Out[∘]= {{8, 0}, {0, -2}}

Problem #2 :

Problem 2:
Consider 3 masses that can slide freely on a circular wire of radius R.
You may wish to label their positions $x_i = R\,\theta_i$, and thus $v_i = x_i' = R\,\theta_i'$,
a) Write down the kinetic energy T and potential energy V for the system in terms of θ_i and θ_i'. Assume all springs have equal k.
b) Using the Lagrange Equations, compute the 3 equations of motion.
c) Derive the T and V matrices.
[You should be alble to get the factors of 2 correctly.]



$$\frac{\partial L}{\partial \mathbf{r}_k} - \frac{\mathrm{d}}{\mathrm{d}t}\frac{\partial L}{\partial \dot{\mathbf{r}}_k} + \sum_{i=1}^{C} \lambda_i \frac{\partial f_i}{\partial \mathbf{r}_k} = 0,$$

In[∘]:=
```
Clear["Global`*"]
```

In[∘]:= $\text{kin} = \frac{1}{2}\left(m1\, R^2\, \theta 1'[t]^2 + m2\, R^2\, \theta 2'[t]^2 + m3\, R^2\, \theta 3'[t]^2\right)$

Out[∘]= $\frac{1}{2}\left(m1\, R^2\, \theta 1'[t]^2 + m2\, R^2\, \theta 2'[t]^2 + m3\, R^2\, \theta 3'[t]^2\right)$

```
In[∘]:= pot = 1/2 k R^2 ((θ1[t] - θ2[t])^2 + (θ2[t] - θ3[t])^2 + (θ3[t] - θ1[t])^2)
```

$$\frac{1}{2} k R^2 \left((\theta 1[t]-\theta 2[t])^2+(\theta 2[t]-\theta 3[t])^2+(-\theta 1[t]+\theta 3[t])^2\right)$$

```
In[∘]:= pot // Expand // Collect[#, k R^2] &
```

$$k R^2 \left(\theta 1[t]^2-\theta 1[t]\times\theta 2[t]+\theta 2[t]^2-\theta 1[t]\times\theta 3[t]-\theta 2[t]\times\theta 3[t]+\theta 3[t]^2\right)$$

```
In[∘]:= lag = kin - pot
```

$$-\frac{1}{2} k R^2 \left((\theta 1[t]-\theta 2[t])^2+(\theta 2[t]-\theta 3[t])^2+(-\theta 1[t]+\theta 3[t])^2\right)+\frac{1}{2}\left(m1 R^2 \theta 1'[t]^2+m2 R^2 \theta 2'[t]^2+m3 R^2 \theta 3'[t]^2\right)$$

```
In[∘]:= D[lag, θ1[t]] // Simplify
```

$$-k R^2 (2 \theta 1[t]-\theta 2[t]-\theta 3[t])$$

```
In[∘]:= D[D[lag, θ1'[t]], t]
```

$$m1 R^2 \theta 1''[t]$$

```
In[∘]:= eq1 = D[lag, θ1[t]] - D[D[lag, θ1'[t]], t] // Simplify
```

$$R^2 (-2 k \theta 1[t]+k \theta 2[t]+k \theta 3[t]-m1 \theta 1''[t])$$

```
In[∘]:= eq2 = D[lag, θ2[t]] - D[D[lag, θ2'[t]], t] // Simplify
```

$$R^2 (k \theta 1[t]-2 k \theta 2[t]+k \theta 3[t]-m2 \theta 2''[t])$$

```
In[∘]:= eq3 = D[lag, θ3[t]] - D[D[lag, θ3'[t]], t] // Simplify
```

$$R^2 (k \theta 1[t]+k \theta 2[t]-2 k \theta 3[t]-m3 \theta 3''[t])$$

```
In[∘]:= eqs = {eq1, eq2, eq3} /. {m1 → m, m2 → m, m3 → m};
        eqs // MatrixForm
```

Out[∘]//MatrixForm=

$$\begin{pmatrix} R^2 (-2 k \theta 1[t]+k \theta 2[t]+k \theta 3[t]-m \theta 1''[t]) \\ R^2 (k \theta 1[t]-2 k \theta 2[t]+k \theta 3[t]-m \theta 2''[t]) \\ R^2 (k \theta 1[t]+k \theta 2[t]-2 k \theta 3[t]-m \theta 3''[t]) \end{pmatrix}$$

```
In[∘]:= -eqs /. {m → 0, R → 1, k → 1} // MatrixForm
```

Out[∘]//MatrixForm=

$$\begin{pmatrix} 2 \theta 1[t]-\theta 2[t]-\theta 3[t] \\ -\theta 1[t]+2 \theta 2[t]-\theta 3[t] \\ -\theta 1[t]-\theta 2[t]+2 \theta 3[t] \end{pmatrix}$$

```
In[∘]:= vmat = {{2, -1, -1}, {-1, 2, -1}, {-1, -1, 2}};
        k vmat // MatrixForm
```

Out[∘]//MatrixForm=

$$\begin{pmatrix} 2k & -k & -k \\ -k & 2k & -k \\ -k & -k & 2k \end{pmatrix}$$

```
In[•]:= -eqs /. {m → 1, R → 1, k → 0} // MatrixForm
```

Out[•]//MatrixForm=

$$\begin{pmatrix} \theta 1''[t] \\ \theta 2''[t] \\ \theta 3''[t] \end{pmatrix}$$

```
In[•]:= tmat = DiagonalMatrix[{1, 1, 1}];
        m tmat // MatrixForm
```

Out[•]//MatrixForm=

$$\begin{pmatrix} m & 0 & 0 \\ 0 & m & 0 \\ 0 & 0 & m \end{pmatrix}$$

```
In[•]:= mat = vmat - w2 tmat;
        mat // MatrixForm
```

Out[•]//MatrixForm=

$$\begin{pmatrix} 2 - w2 & -1 & -1 \\ -1 & 2 - w2 & -1 \\ -1 & -1 & 2 - w2 \end{pmatrix}$$

```
In[•]:= Det[mat]
```

Out[•]= $-9\,w2 + 6\,w2^2 - w2^3$

```
In[•]:= Solve[Det[mat] == 0, w2]
```

Out[•]= {{w2 → 0}, {w2 → 3}, {w2 → 3}}

Problem #3

```
In[•]:= Clear["Global`*"]
```

In[•]:= term = $\sqrt{\frac{x'[y]^2 + 1}{y}}$

Out[•]= $\sqrt{\frac{1 + x'[y]^2}{y}}$

```
In[•]:= D[term, x[y]]
```

Out[•]= 0

```
In[•]:= D[D[term, x'[y]], y]
```

Out[•]=

$$-\frac{x'[y]}{y^2\sqrt{\frac{1+x'[y]^2}{y}}} + \frac{x''[y]}{y\sqrt{\frac{1+x'[y]^2}{y}}} - \frac{x'[y]\left(-\frac{1+x'[y]^2}{y^2} + \frac{2\,x'[y]\,x''[y]}{y}\right)}{2\,y\left(\frac{1+x'[y]^2}{y}\right)^{3/2}}$$

In[∘]:= `tmp2 = D[term, x'[y]] // PowerExpand`

Out[∘]= $\frac{x'[y]}{\sqrt{y}\ \sqrt{1 + x'[y]^2}}$

In[∘]:= `eq = tmp2^2 == 1/(2 a)`

Out[∘]= $\frac{x'[y]^2}{y\,(1 + x'[y]^2)} == \frac{1}{2\,a}$

In[∘]:= `sol = Solve[eq, x'[y]]`

Out[∘]= $\left\{\left\{x'[y] \to -\frac{\sqrt{y}}{\sqrt{2\,a - y}}\right\}, \left\{x'[y] \to \frac{\sqrt{y}}{\sqrt{2\,a - y}}\right\}\right\}$

In[∘]:= `tmp4 = x'[y] /. sol[[2]]`

Out[∘]= $\frac{\sqrt{y}}{\sqrt{2\,a - y}}$

In[∘]:= `jacobian = D[a (1 - Cos[θ]), θ]`

Out[∘]= $a\,\mathrm{Sin}[\theta]$

In[∘]:= `tmp5 = tmp4 jacobian /. {y → a (1 - Cos[θ])}`

Out[∘]= $\frac{a\,\sqrt{a\,(1 - \mathrm{Cos}[\theta])}\ \mathrm{Sin}[\theta]}{\sqrt{2\,a - a\,(1 - \mathrm{Cos}[\theta])}}$

In[∘]:= `tmp6 = Sqrt[tmp5^2] // FullSimplify // PowerExpand`

Out[∘]= $2\,a\,\mathrm{Sin}\left[\frac{\theta}{2}\right]^2$

In[∘]:= `tmp7 = tmp6 // TrigExpand // FullSimplify`

Out[∘]= $a - a\,\mathrm{Cos}[\theta]$

In[∘]:= `tmp8 = Integrate[tmp7, θ]`

Out[∘]= $a\,\theta - a\,\mathrm{Sin}[\theta]$

In[∘]:=
```
y[θ_] = -tmp7
x[θ_] = tmp8
```

Out[∘]= $-a + a\,\mathrm{Cos}[\theta]$

Out[∘]= $a\,\theta - a\,\mathrm{Sin}[\theta]$

In[∘]:=
```
ParametricPlot[{x[θ], y[θ]} /. {a → 1}, {θ, 0, 6 π}]
```

Out[∘]=



Problem #4 :

Problem 4:

Consider a mass m_1 at position x hanging from a pulley of mass m_2 at angle θ and moment of inertia $I = \alpha\, m_2 R^2$.
a) Using Lagrange multipliers, compute the equation of motion for x and θ with constraint equation x - R θ =0.
b) Solve for the acceleration x.



In[∘]:=
```
Clear["Global`*"]
```

In[∘]:=
```
kin = 1/2 m1 x'[t]^2 + 1/2 α m2 R^2 θ'[t]^2
```

Out[∘]= $\frac{1}{2}\, m1\, x'[t]^2 + \frac{1}{2}\, m2\, R^2\, \alpha\, \theta'[t]^2$

In[∘]:=
```
pot = - m1 g x[t]
```

Out[∘]= $-g\, m1\, x[t]$

In[∘]:=
```
lag = kin - pot
```

Out[∘]= $g\, m1\, x[t] + \frac{1}{2}\, m1\, x'[t]^2 + \frac{1}{2}\, m2\, R^2\, \alpha\, \theta'[t]^2$

In[∘]:=
```
constraint = x[t] - R θ[t]
```

Out[∘]= $x[t] - R\, \theta[t]$

In[∘]:=
```
{D[constraint, x[t]], D[constraint, θ[t]]}
```

Out[∘]= {1, -R}

In[∘]:=
```
eqx = D[lag, x[t]] - D[D[lag, x'[t]], t] + λ D[constraint, x[t]] == 0 // Simplify
```

Out[∘]= $g\, m1 + \lambda == m1\, x''[t]$

In[∘]:=
```
eqθ = D[lag, θ[t]] - D[D[lag, θ'[t]], t] + λ D[constraint, θ[t]] == 0 // Simplify
```

Out[∘]= $R\left(\lambda + m2\, R\, \alpha\, \theta''[t]\right) == 0$

In[∘]:=
```
eqF = D[constraint, {t, 2}] == 0
```

Out[∘]= $x''[t] - R\, \theta''[t] == 0$

In[∘]:=
```
sol = Solve[{eqx, eqθ, eqF}, {x''[t], θ''[t], λ}] // Simplify
```

Out[∘]= $\left\{\left\{x''[t] \to \frac{g\, m1}{m1 + m2\, \alpha},\ \theta''[t] \to \frac{g\, m1}{m1\, R + m2\, R\, \alpha},\ \lambda \to -\frac{g\, m1\, m2\, \alpha}{m1 + m2\, \alpha}\right\}\right\}$

In[∘]:=
```
sol /. {α → 0}
```

Out[∘]= $\left\{\left\{x''[t] \to g,\ \theta''[t] \to \frac{g}{R},\ \lambda \to 0\right\}\right\}$

Problem #5 :

Problem 5: (Delta Functions)
Evaluate the delta functions

a) $\int_0^\infty \delta(x-2)$

b) $\int_0^\infty \delta(2x-2)$

c) $\int_0^\infty (5x-2)\ \delta(x-2)$

d) $\int_0^\infty \phi(x)\ \delta(x^2-a^2)$

e) $\int_{-\pi/2}^{\pi/2} \cos(x)\ \delta(\sin(x))$

f) $\int_0^{2\pi} \delta(\cos(x))$

g) $\int_0^\infty \phi(x)\ \delta'(x)$

In[∘]:=
```
Clear["Global`*"]
```

In[∘]:=
```
Integrate[DiracDelta[x - 2], {x, 0, Infinity}]
```

Out[∘]= 1

In[∘]:=
```
Integrate[DiracDelta[2 x - 2], {x, 0, Infinity}]
```

Out[∘]= $\frac{1}{2}$

In[∘]:=
```
Integrate[(5 x - 2) DiracDelta[x - 2], {x, 0, Infinity}]
```

Out[∘]= 8

In[∘]:=
```
Integrate[ϕ[x] DiracDelta[(x^2 - a^2)], {x, 0, Infinity}, Assumptions → {a > 0}]
```

Out[∘]= $\frac{\phi[a]}{2\, a}$

```
In[◦]:= Integrate[Cos[x] DiracDelta[Sin[x]], {x, -π/2, π/2}]
Out[◦]= 1

In[◦]:= Integrate[DiracDelta[Cos[x]], {x, 0, 2 π}]
Out[◦]= 2

In[◦]:= Integrate[DiracDelta[Cos[x]], {x, 0, 4 π}]
Out[◦]= 4

In[◦]:= Integrate[ϕ[x] DiracDelta'[x], {x, -Infinity, Infinity}]
Out[◦]= -ϕ'[0]
```

Problem #6 :

```
In[◦]:= Clear["Global`*"]

In[◦]:= Integrate[g[x] × h[y] DiracDelta[x + y - z],
          {x, 0, 1}, {y, 0, 1}, Assumptions → {1 > z > 0, z > x}] // Expand
Out[◦]= Integrate[g[x] × h[-x + z], {x, 0, 1}, Assumptions → {1 > z > 0, z > x}]

In[◦]:= g[x_] = Exp[-x] - Exp[-2 x];
        h[y_] = Exp[-y];

In[◦]:= Integrate[g[x] × h[y] DiracDelta[x + y - z], {x, 0, 1}, {y, 0, 1}, Assumptions → {1 > z > 0}] // Expand
```

Out[◦]= $e^{-2 z} - e^{-z} + e^{-z} z$

```
In[◦]:= Integrate[g[x] × h[z - x] , {x, 0, z}, Assumptions → {1 > z > 0}] // Expand
```

Out[◦]= $e^{-2 z} - e^{-z} + e^{-z} z$

```
In[◦]:= Integrate[g[z - y] × h[y] , {y, 0, z}, Assumptions → {1 > z > 0}] // Expand
```

Out[◦]= $e^{-2 z} - e^{-z} + e^{-z} z$

Problem #7

```
In[◦]:= term[n_, x_, y_] := Sin[n π x] Exp[- n π y]
```

```
In[∘]:= Plot3D[term[1, x, y], {x, 0, 1}, {y, 0, 1}, PlotRange → All]
```

Out[∘]=




```
In[∘]:= c[n_] = Integrate[ Sin[n π x] , {x, 0, 1}]
```

Out[∘]= $\frac{1 - \text{Cos}[n\,\pi]}{n\,\pi}$

```
In[∘]:= Limit[c[n], n → 0]
Out[∘]= 0

In[∘]:= c[0] = 0
Out[∘]= 0

In[∘]:= sum[m_, x_, y_] := Sum[ c[n] × term[n, x, y], {n, 0, m}]

In[∘]:= Plot3D[sum[30, x, y], {x, 0, 1}, {y, 0, 1}, PlotRange → All]
```

Out[∘]=